In[2]:=
```
SetDirectory["~/KappaLib/"];
<< kappaLib-1.2.m
<< helper.m
```

```
Loading KappaLib v1.2

Loading helper.m..
```

In this notebook we define a 1-parameter class of mediums:

1) each kappa is algebraically decomposable

2) beta^2 -alpha gamma = 0

3) the Fresnel polynomial always has four linear factors

- **Define medium**

In[5]:=

$$\texttt{kappa = emMatrixToKappa}\left[\begin{pmatrix} 1 & 1 & 1 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 2 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -k65 & -1 \\ 1 & 1 & 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 & k65 & 0 \end{pmatrix}\right];$$

In[6]:=
```
(* kappa is invertible *)
FullSimplify[emDet[kappa]]

(* kappa has no axion component *)
Simplify[emTrace[kappa]]

(* Since this does not simplify to {0}, kappa has a skewon component *)
Union[Flatten[kappa - emPoincare[kappa]]]

(* Since this does not simplify to {0}, kappa has a principal component *)
Union[Flatten[kappa + emPoincare[kappa]]]
```

Out[6]= 1

Out[7]= 0

Out[8]= {-3, -2, -1, 0, 1, 2, 3, -k65, 1 + k65}

Out[9]= {-2, -1, 0, 1, 1 - k65, -1 + k65, -k65, k65}

- **Kappa depends on 1-variable**

In[10]:=
```
Variables[kappa]
```

Out[10]= {k65}

- **Define constants alpha, beta, gamma and bivectors A and B:**

In[11]:=
```
alpha = 1;
beta = 1;
gamma = 1;
rho = 1;
```

$$\text{Abivector} = \begin{pmatrix} 0 & -\frac{1}{\sqrt{2}} & -\frac{1}{\sqrt{2}} & 0 \\ \frac{1}{\sqrt{2}} & 0 & 0 & 0 \\ \frac{1}{\sqrt{2}} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix};$$

```
Bbivector = Abivector;
```

- **Note these satisfy beta^2-alpha gamma = 0**

In[17]:=
```
beta ^ 2 - alpha gamma
```

Out[17]= 0

- **Verify that kappa is algebraically decomposable**

In[18]:=
```
LHS = alpha emIdentityKappa[] +
   beta (kappa + emPoincare[kappa]) + gamma emCompose[emPoincare[kappa], kappa];
RHS = emBiProduct[rho, Abivector, Bbivector] + emBiProduct[rho, Bbivector, Abivector];
Union[Simplify[Flatten[(LHS - RHS)]]]
```

Out[20]= {0}

- **Compute Fresnel polynomial**

In[21]:=
```
coords = {xi0, xi1, xi2, xi3};
fresnel = FullSimplify[emKappaToFresnel[kappa, coords]];
```

- **Fresnel polynomial is product of linear factor and 3rd order factor**

In[23]:=
```
fresnel
```

Out[23]= $-(-1 + k65)\ xi0\ (xi1 - xi3)^2\ (xi1 + xi3)$

- **Note:**
  - The Fresnel polynomial has four linear factors
  - When k65 = 1, the Fresnel polynomial is the zero polynomial

---

Show that

D (kappa + beta Id) = 1/2 trace (rho bar(D) otimes D) A + B

has a solution for D

In[24]:=
```
(*
 * If D=1/2 D^ij d/dx^i /\ d/dx^j is a bivector we represent
 * the coefficients by the anti-symmetric matrix with coefficients
 * of (D^ij)_ij. If kappa is an antisymmetric (2,2)-tensor,
 * then this routine returns coefficients of bivector D (kappa).
 *)
contract[biv_,kappa_] := Table[
 1/2 Sum[biv[[i]][[j]]emReadNormal[kappa,a,b,i,j]
      ,
      {i, 1, 4},{j, 1, 4}
    ]
 ,
 {a, 1, 4}, {b, 1, 4}
]
```

In[25]:=
```
Dbivector = {{0, 0, -1/Sqrt[2], 0}, {0, 0, 0, 0}, {1/Sqrt[2], 0, 0, 0}, {0, 0, 0, 0}};
dLHS = contract[Dbivector, kappa + beta emIdentityKappa[]];
dRHS = 1 / 2 emTrace[emBiProduct[rho, Dbivector, Dbivector]] Abivector + Bbivector;
deqs = Union[Flatten[Simplify[dLHS - dRHS]]]
```

$$\text{Dbivector} = \begin{pmatrix} 0 & 0 & -\frac{1}{\sqrt{2}} & 0 \\ 0 & 0 & 0 & 0 \\ \frac{1}{\sqrt{2}} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{pmatrix};$$

Out[28]= {0}